
    hnc-csv - CSV Decoder/Encoder

Decoding
Whole CSV binary documents can be decoded with decode/1,2.
decode/1 assumes default RFC4180-style
options, that is:
	Fields are separated by commas.
	Fields are optionally enclosed in double quotes.
	Double quotes in enclosed fields are quoted by another double quote.

decode/2 allows using custom options:
#{separator => Separator, % any byte except $\r or $\n (defaul $,)
  enclosure => Enclosure, % 'undefined' or any byte except $\r or $\n (default $")
  quote     => Quote}     % 'undefined', 'enclosure', or any byte except $\r or $\n (defaults 'enclosure')
Restrictions for option combinations:
	If Enclosure is undefined (ie, no enclosing), Quote must be either enclosure or undefined.
	If Enclosure is not undefined, Quote must also not be undefined.
	If Enclosure is not undefined, it must not be the same as Separator.

Lines are separated by \r, \n or \r\n. Empty lines are ignored by the decoder.
The result of decoding is a list of CSV lines, which are lists of CSV fields,
which are in turn binaries representing the field values on the respective line.
Example
Assume the following CSV data:
a,b,c
"d,d","e""e","f
f"
In an Erlang binary, this will look like:
1> CsvBinary = <<"a,b,c\r\n\"d,d\",\"e\"\"e\",\"f\r\nf\"\r\n">>.
<<"a,b,c\r\n\"d,d\",\"e\"\"e\",\"f\r\nf\"\r\n">>
Decoded with decode/1, this will become:
2> hnc_csv:decode(CsvBinary).
[[<<"a">>,<<"b">>,<<"c">>],
 [<<"d,d">>,<<"e\"e">>,<<"f\r\nf">>]]
Higher Order Functions for Decoding
hnc_csv provides the functions decode_fold/3,4, decode_filter/2,3,
decode_map/2,3, decode_filtermap/2,3 and decode_foreach/2,3 which
allow decoding and processing decoded lines in one operation, much
like the lists functions foldl/3, filter/2, map/2, filtermap/2
and foreach/2.
In fact, decode/1,2 is implemented via decode_fold/3,4.
Providers
The decode family of functions accepts both a raw binary as well as a
Provider that delivers chunks of raw binary. When given a raw binary,
it is converted into a binary provider for further processing.
A provider is a 0-arity function which, when called, returns either a
tuple where the first element is a chunk of binary data and the second
is a new provider function for the next chunk of data, or the atom
end_of_data to indicate that the provider has delivered all data.
Providers can be implemented stateless of stateful, usually depending
on the characteristics of the underlying data source.
A stateless provider does not change and is not susceptible to external
changes to the state of the underlying data source.
A stateful provider on the other hand may change or be susceptible to
changes to the state of the underlying data source or both. It is recommended
to not (re-)use stateful providers or their underlying data source before, while
or after being used in decoding functions, except for any necessary setup before or
cleanup after being used.
hnc_csv comes with two convenience functions, get_binary_provider/1,2
(stateless) and get_file_provider/1,2 (stateful) which return providers for
binaries or files, respectively.
Example
The following is an implementation of a (stateless) custom provider which delivers
data taken from a given list of binaries:
-module(example_provider).
-export([get_list_provider/1]).

get_list_provider(L) ->
    fun() -> list_provider(L) end.

list_provider([]) ->
    end_of_data;
list_provider([Bin|More]) when is_binary(Bin) ->
    {Bin, fun() -> list_provider(More) end}.
	get_list_provider/1 creates the initial provider, which is a call
to list_provider/1 wrapped in a 0-arity function.
	list_provider/1 is the actual implementation of the provider, which
returns either end_of_data when the list given as argument is exhausted,
or otherwise a tuple with the head element of the list as first
and a call to itself with the tail of the list wrapped in a 0-arity
function as second element.

This provider can then be used as follows, for example to count the lines
and fields in the CSV data which the provider delivers:
1> Provider = example_provider:get_list_provider([<<"a,b">>, <<",c\r">>,
                                                  <<"\nd,">>, <<"e,f">>,
                                                  <<"\r\n">>]).
#Fun<example_provider.0.64990923>
2> hnc_csv:decode_fold(Provider,
                       fun(Line, {LCnt, FCnt}) -> {LCnt+1, FCnt+length(Line)} end,
                       {0, 0}).
{2,6}
Advanced Usage
For more complex scenarios than what the built-in functions provide
for, the functions decode_init/0,1,2, decode_next_line/1 and
decode_flush/1 can be used together to decode and process CSV
documents incrementally.
	decode_init/0,1,2 creates a decoder state to be used in the
other functions listed above.
	decode_next_line/1 decodes and returns the next line, together with
an updated state. If the data in the provider backing the state is exhausted,
the atom end_of_data is returned instead of a line.
	decode_flush/1 returns all as by then unread lines in the given state.

In fact, decode_fold/4 is implemented using those functions.
Encoding
CSV documents can be encoded with encode/1,2.
encode/1 assumes default RFC4180-style
options, that is:
	Fields are separated by commas
	Fields are optionally enclosed in double quotes
	Double quotes in enclosed fields are quoted by another double quote
	Lines are separated by \r\n

encode/2 allows using custom options:
#{separator   => Separator, % any byte except $\r and $\n (default $,)
  enclosure   => Enclosure, % 'undefined' or any byte except $\r or $\n (default $")
  quote       => Quote,     % 'undefined', 'enclosure', or any byte except $\r or $\n (default 'enclosure')
  enclose     => Enclose,   % 'optional' (default), 'never' or 'always'
  end_of_line => EndOfLine} % `<<"\r\n">> (default), <<"\n">> or <<"\r">>
Restrictions for option combinations:
	If Enclose is never (ie, no enclosing), Enclosure must be undefined and Quote must be undefined or enclosure.
	If Enclose is optional or always, Enclosure and Quote must not be undefined.
	If Enclosure is not undefined, it must not be the same as Separator.

The input for encoding is a list of CSV lines, which are in turn lists of CSV fields,
which are in turn binaries representing the field values.
The result is a CSV binary document consisting of the given CSV lines, in turn consisting of the given CSV fields of a line.
Example
Assume the following CSV structure:
1> Csv = [[<<"a">>,<<"b">>,<<"c">>],
          [<<"d,d">>,<<"e\"e">>,<<"f\r\nf">>]].
Encoded with encode/1, this will become:
2> hnc_csv:encode(Csv).
<<"a,b,c\r\n"
  "\"d,d\",\"e\"\"e\",\"f\r\nf\"\r\n">>
Authors
	Maria Scott (Maria-12648430)

    LICENSE


Copyright (c) 2024-2025, Maria Scott <maria-12648430@hnc-agency.org>
Copyright (c) 2024-2025, Jan Uhlig <juhlig@hnc-agency.org>

Permission to use, copy, modify, and/or distribute this software for any
purpose with or without fee is hereby granted, provided that the above
copyright notice and this permission notice appear in all copies.

THE SOFTWARE IS PROVIDED "AS IS" AND THE AUTHOR DISCLAIMS ALL WARRANTIES
WITH REGARD TO THIS SOFTWARE INCLUDING ALL IMPLIED WARRANTIES OF
MERCHANTABILITY AND FITNESS. IN NO EVENT SHALL THE AUTHOR BE LIABLE FOR
ANY SPECIAL, DIRECT, INDIRECT, OR CONSEQUENTIAL DAMAGES OR ANY DAMAGES
WHATSOEVER RESULTING FROM LOSS OF USE, DATA OR PROFITS, WHETHER IN AN
ACTION OF CONTRACT, NEGLIGENCE OR OTHER TORTIOUS ACTION, ARISING OUT OF
OR IN CONNECTION WITH THE USE OR PERFORMANCE OF THIS SOFTWARE.

          -spec decode(ProviderOrRawData) -> Lines
                when
                    ProviderOrRawData :: Provider | RawData,
                    Provider :: provider(),
                    RawData :: data(),
                    Lines :: [Line],
                    Line :: csv_line().


      


Equivalent to decode(ProviderOrRawData, default_decode_options()).

          -spec decode(ProviderOrRawData, Options) -> Lines
                when
                    ProviderOrRawData :: Provider | RawData,
                    Provider :: provider(),
                    RawData :: data(),
                    Options :: decode_options(),
                    Lines :: [Line],
                    Line :: csv_line().


      


Decodes the raw CSV document in RawData or provided by Provider, using the given Options, into a CSV data structure.
The return value is a list of CSV lines, which in turn are lists of CSV fields, which in turn are binaries representing the CSV field values.

          -spec decode_filter(ProviderOrRawData, Fun) -> Lines
                       when
                           ProviderOrRawData :: Provider | RawData,
                           Provider :: provider(),

                           Fun :: fun((Line) -> boolean()),
                           Line :: csv_line(),
                           Lines :: [csv_line()].


      


Equivalent to decode_filter(ProviderOrRawData, default_decode_options(), Fun).

          -spec decode_filter(ProviderOrRawData, Options, Fun) -> Lines
                       when
                           ProviderOrRawData :: Provider | RawData,
                           Provider :: provider(),
                           RawData :: data(),
                           Options :: decode_options(),
                           Fun :: fun((Line) -> boolean()),
                           Line :: csv_line(),
                           Lines :: [csv_line()].


      


Successively calls the given function Fun with each CSV line decoded from the raw binary data in RawData or provided by Provider as argument.
Returns a list of CSV lines for which Fun returned true.

                          when
                              ProviderOrRawData :: Provider | RawData,
                              Provider :: provider(),
                              RawData :: data(),
                              Fun :: fun((Line) -> boolean() | {true, Mapped}),

                              Mapped :: term(),
                              Result :: [Line | Mapped].

          -spec decode_filtermap(ProviderOrRawData, Options, Fun) -> Result
                          when
                              ProviderOrRawData :: Provider | RawData,
                              Provider :: provider(),
                              RawData :: data(),
                              Options :: decode_options(),
                              Fun :: fun((Line) -> boolean() | {true, Mapped}),
                              Line :: csv_line(),
                              Mapped :: term(),
                              Result :: [Line | Mapped].


      


Combines the functionality of decode_filter and decode_map.
If Fun returns true or false for a CSV line given to it, the CSV line is kept or discarded from the result list (filtering). If it returns {true, Mapped}, Mapped is inserted in the result list at the respective position (mapping).

          -spec decode_flush(State) -> Result when State :: state(), Result :: [Line], Line :: csv_line().


      


Reads and decodes any data remaining in the given CSV decoder state.
Returns a list of the remaining CSV lines.

          -spec decode_fold(ProviderOrRawData, Fun, Acc0) -> Acc1
                     when
                         ProviderOrRawData :: Provider | RawData,
                         Provider :: provider(),
                         RawData :: data(),
                         Fun :: fun((Line, AccIn) -> AccOut),
                         Line :: csv_line(),
                         Acc0 :: term(),
                         Acc1 :: term(),
                         AccIn :: term(),
                         AccOut :: term().


      


Equivalent to decode_fold(ProviderOrRawData, default_decode_options(), Fun, Acc0).

                     when
                         ProviderOrRawData :: Provider | RawData,
                         Provider :: provider(),
                         RawData :: data(),
                         Options :: decode_options(),
                         Fun :: fun((Line, AccIn) -> AccOut),
                         Line :: csv_line(),
                         Acc0 :: term(),
                         Acc1 :: term(),
                         AccIn :: term(),
                         AccOut :: term().


      


Successively calls the given function Fun with each CSV line decoded from the raw binary data in RawData or provided by Provider as first and an accumulator as second arguments.
Returns the final accumulator.

          -spec decode_foreach(ProviderOrRawData, Fun) -> ok
                        when
                            ProviderOrRawData :: Provider | RawData,
                            Provider :: provider(),
                            RawData :: data(),
                            Fun :: fun((Line) -> _),
                            Line :: csv_line().


      


Equivalent to decode_foreach(ProviderOrRawData, default_decode_options(), Fun).

          -spec decode_foreach(ProviderOrRawData, Options, Fun) -> ok
                        when
                            ProviderOrRawData :: Provider | RawData,
                            Provider :: provider(),
                            RawData :: data(),
                            Options :: decode_options(),
                            Fun :: fun((Line) -> _),
                            Line :: csv_line().


      


Successively calls the given function Fun with each CSV line decoded from the raw binary data in RawData or provided by Provider as argument.

          -spec decode_init(ProviderOrRawData) -> State
                     when
                         ProviderOrRawData :: Provider | RawData,
                         Provider :: provider(),
                         RawData :: data(),
                         State :: state().


      


Equivalent to decode_init(ProviderOrRawData, default_decode_options()).

          -spec decode_init(ProviderOrRawData, Options) -> State
                     when
                         ProviderOrRawData :: Provider | RawData,
                         Provider :: provider(),
                         RawData :: data(),
                         Options :: decode_options(),
                         State :: state().


      


Creates a CSV decoder state to incrementally decode the raw CSV document in RawData or provided by Provider, using the given Options.
The return value can be used as argument to the function decode_next_line/1 to obtain CSV lines one by one.
See also: decode_next_line/1.

          -spec decode_map(ProviderOrRawData, Fun) -> Result
                    when
                        ProviderOrRawData :: Provider | RawData,
                        Provider :: provider(),
                        RawData :: data(),
                        Fun :: fun((Line) -> Mapped),
                        Line :: csv_line(),
                        Mapped :: term(),
                        Result :: [Mapped].


      


Equivalent to decode_map(ProviderOrRawData, default_decode_options(), Fun).

          -spec decode_map(ProviderOrRawData, Options, Fun) -> Result
                    when
                        ProviderOrRawData :: Provider | RawData,
                        Provider :: provider(),
                        RawData :: data(),
                        Options :: decode_options(),
                        Fun :: fun((Line) -> Mapped),
                        Line :: csv_line(),
                        Mapped :: term(),
                        Result :: [Mapped].


      


Successively calls the given function Fun with each CSV line decoded from the raw binary data in RawData or provided by Provider as argument.
Returns a list of the values returned by Fun.

          -spec decode_next_line(State0) -> Result
                          when
                              State0 :: state(),
                              Result :: end_of_data | {Line, State1},
                              Line :: csv_line(),
                              State1 :: state().


      


Decodes and returns the next CSV line in the given decoder state, together with an updated state which can be used for further incremental decoding.
If the provider of the decoder state is exhausted, the atom end_of_data is returned instead of a line.

          -spec default_decode_options() -> decode_options().


      


Returns the default decode options.
	separator: $,
	enclosure: $"
	quote: enclosure

          -spec default_encode_options() -> encode_options().


      


Returns the default encode options.
	separator: $,
	enclosure: $"
	quote: enclosure
	enclose: optional
	end_of_line: <<"\r\n">>

          -spec encode(Lines) -> RawData when Lines :: [Line], Line :: csv_line(), RawData :: data().


      


Equivalent to encode(Lines, default_encode_options()).

  



  
    
      
    
    
      encode(Lines, Options)



        
          
        

    

  


  

      

          -spec encode(Lines, Options) -> RawData
                when Lines :: [Line], Line :: csv_line(), Options :: encode_options(), RawData :: data().


      


Encodes the given CSV data structure into a CSV binary, using the given Options.

          -spec flush_provider(Provider) -> RawData when Provider :: provider(), RawData :: data().


      


Flushes and returns all data remaining in the given Provider.

          -spec get_binary_provider(Binary) -> Provider when Binary :: binary(), Provider :: provider().


      


Equivalent to get_binary_provider(Binary, 1024).

          -spec get_binary_provider(Binary, ChunkSize) -> Provider
                             when Binary :: binary(), ChunkSize :: pos_integer(), Provider :: provider().


      


Creates a data provider from the given Binary to supply data in chunks of the given ChunkSize.
This provider can be used in decode_init/1,2. decode_fold/3,4, decode_foreach/2,3, decode_filter/2,3, decode_map/2,3 and decode_filtermap/2,3.

          -spec get_file_provider(IoDevice) -> Provider
                           when IoDevice :: file:io_device() | io:device(), Provider :: provider().


      


Equivalent to get_file_provider(IoDevice, 1024).

          -spec get_file_provider(IoDevice, ChunkSize) -> Provider
                           when
                               IoDevice :: file:io_device() | io:device(),
                               ChunkSize :: pos_integer(),
                               Provider :: provider().


      


Creates a data provider from an open file given in IoDevice to supply data in chunks of the given ChunkSize.
The file must have been opened with modes read and binary.
Note that this provider is stateful, because the underlying IoDevice is stateful. This means that reading from the provider changes the state of the IoDevice (namely, advancing the position), and is itself affected by any external changes to the state of the IoDevice.
When the position of the IoDevice is at eof, the provider is exhausted.
The IoDevice is not closed implicitly by the provider, instead the code using this provider is responsible for closing it.
This provider can be used in decode_init/1,2. decode_fold/3,4, decode_foreach/2,3, decode_filter/2,3, decode_map/2,3 and decode_filtermap/2,3.

